Observational data
For each season format the station data in CPT format, i.e., with columns as stations, and rows as years. Start at the beginning of the climatological period and continue to as up to date as possible. For missing data leave the cell blank.

Observations

	
	A
	B
	C
	D

	1
	Station
	Station A
	Station B
	Station C

	2
	Latitude
	6.25
	8.75
	11.25

	3
	Longitude
	1.25
	1.25
	1.25

	4
	1961
	80.4
	189.4
	191.9

	5
	1962
	113.2
	212.4
	236.7

	6
	1963
	231.7
	304.2
	239.6

	7
	1964
	72.8
	195.7
	246.3


Calculate the terciles using the PERCENTILE function:

=PERCENTILE(B$4:B$33,1/3)

=PERCENTILE(B$4:B$33,2/3)

The climatological period represents the first 30 years, which, in the example, are in rows 4 to 33. The second argument represents the percentile; we need the 67th for the upper tercile, and the 33rd for the lower. Use the $ signs so you can copy the formula for the upper tercile one row down to get the formula for the lower tercile, and the you only need to change 2/3 to 1/3.

As a check, use COUNTIF to confirm that each category occurs 33% of the time:

=COUNTIF(B$4:B$33,"<"&B52)/COUNT(B$4:B$33)

=COUNTIF(B$4:B$33,">="&B53)/COUNT(B$4:B$33)

(assuming that the lower tercile is in row 52, and the upper in 53). We do not need to count for the normal category if the other two categories occur the correct number of times.

If there are missing values the category checks may not come out at exactly 33%, but should typically be within about 1%.

Verifying categories

Using the terciles we can identify what were the verifying categories for the years we have forecasts, by using a nested =IF() statement:

=IF(Observations!B44<=Observations!B$52,"B", IF(Observations!B44<=Observations!B$53,"N","A"))

The first condition checks whether the observation for 2001 (in row 44 in the example) is less than or equal to the lower tercile, returning “B” if so. If not, the next condition checks whether the same observation is less than or equal to the upper tercile, returning “N” if so and “A” if not. Using $ symbols to point to the tercile lines, the formula can be copied to get the verifying categories for all years and stations. 

Categories

	
	A
	B
	C
	D

	1
	Station
	Station A
	Station B
	Station C

	2
	Latitude
	6.25
	8.75
	11.25

	3
	Longitude
	1.25
	1.25
	1.25

	4
	2001
	B
	B
	B

	5
	2002
	N
	N
	N

	6
	2003
	A
	A
	A

	7
	2004
	A
	N
	B


If you have missing values during the verification period, and have left these blank (which we need to do for the climatological period in order to calculate the terciles) then unfortunately Excel is going to interpret them as zeroes and almost certainly classify them as Below-normal. That error can be avoided by entering =NA() for missing observations for the years during the verification period, but this inconsistency is far from desirable, and so it would be best to have a separate worksheet for the verifying observations. However, many of the following procedures will need to be modified to deal with the missing values.
Forecast data

For each station enter the corresponding forecasts in a similar format to the observations, but with a separate table for each category, as shown in the Forecasts worksheet below. If we have lots of stations, entering the data may be prone to errors. It is safer to enter the data by regions. Start by assigning each station to a region. The region numbers are arbitrary since we are using them only as a means to obtain the station forecasts; we are not going to verify the forecasts by region, and so if does not matter if the regions differ from year to year.

Forecasts
	
	A
	B
	C
	D

	1
	B
	
	
	

	2
	Station
	Station A
	Station B
	Station C

	3
	Latitude
	6.25
	8.75
	11.25

	4
	Longitude
	1.25
	1.25
	1.25

	5
	2001
	25%
	25%
	25%

	6
	2002
	25%
	25%
	40%

	7
	2003
	25%
	25%
	25%

	8
	2004
	30%
	20%
	30%

	9
	
	
	
	

	10
	N
	
	
	

	11
	Station
	Station A
	Station B
	Station C

	12
	Latitude
	6.25
	8.75
	11.25

	13
	Longitude
	1.25
	1.25
	1.25

	14
	2001
	35%
	35%
	35%

	15
	2002
	35%
	35%
	35%

	16
	2003
	35%
	35%
	35%

	17
	2004
	40%
	35%
	45%

	18
	
	
	
	

	19
	A
	
	
	

	20
	Station
	Station A
	Station B
	Station C

	21
	Latitude
	6.25
	8.75
	11.25

	22
	Longitude
	1.25
	1.25
	1.25

	23
	2001
	40%
	40%
	40%

	24
	2002
	40%
	40%
	25%

	25
	2003
	40%
	40%
	40%

	26
	2004
	30%
	45%
	25%


Regions

	
	A
	B
	C
	D

	1
	Station
	Station A
	Station B
	Station C

	2
	Latitude
	6.25
	8.75
	11.25

	3
	Longitude
	1.25
	1.25
	1.25

	4
	1961
	I
	I
	I

	5
	1962
	I
	I
	II

	6
	1963
	I
	I
	I

	7
	1964
	I
	II
	III


Enter the forecasts for each region in worksheet Probabilities. Keep the blank lines in rows 3 and 4 so that the years for each category will be in the same rows as for the stations eventually. 

Probabilities
	
	A
	B
	C
	D

	1
	B
	
	
	

	2
	Region
	I
	II
	III

	3
	
	
	
	

	4
	
	
	
	

	5
	2001
	25%
	35%
	40%

	6
	2002
	25%
	40%
	35%

	7
	2003
	25%
	40%
	35%

	8
	2004
	30%
	20%
	30%

	9
	
	
	
	

	10
	N
	
	
	

	11
	Region
	I
	II
	III

	12
	
	
	
	

	13
	
	
	
	

	14
	2001
	35%
	35%
	35%

	15
	2002
	35%
	35%
	40%

	16
	2003
	35%
	35%
	45%

	17
	2004
	40%
	35%
	45%

	18
	
	
	
	

	19
	A
	
	
	

	20
	Region
	I
	II
	III

	21
	
	
	
	

	22
	
	
	
	

	23
	2001
	40%
	25%
	25%

	24
	2002
	40%
	25%
	25%

	25
	2003
	40%
	25%
	20%

	26
	2004
	30%
	45%
	25%


To check whether there are any errors, it may be worth calculating the probabilities for Above as 100 – the sum of the probabilities for the other categories. For example, A17 should be =100-A10-A3.

Now use HLOOKUP to pull the forecasts for each station from the table of forecasts for the regions. In Forecasts, B5 becomes:
=HLOOKUP(Regions!B4,Probabilities!$B$2:$D$6,Forecasts!$A5-1997)

(change the reference to the Probabilities table depending on the number of stations and years of forecasts). The function tells Excel to lookup the region number given in Regions!B4 in the table of probabilities in Probabilities!$B$2:$D$6, and specifically to identify which column the region number is in. The table of probabilities that the HLOOKUP function is working with is highlighted below

	
	A
	B
	C
	D

	1
	B
	
	
	

	2
	Region
	I
	II
	III

	3
	
	
	
	

	4
	
	
	
	

	5
	2001
	25%
	35%
	40%

	6
	2002
	25%
	40%
	35%

	7
	2003
	25%
	40%
	35%

	8
	2004
	30%
	20%
	30%


The last argument tells Excel to take the value in the corresponding row of the table; for year 2001 we need the probabilities in row 5 of the spreadsheet, which is row 4 of the table set in the function call. Rather than entering 4 as the last argument, we can generalize to use the same formula for the other years: for 2002 we need the probabilities in the 5th row of the table, and so we can set the row required by subtracting 1997 from the year. Now we can copy B4 of Forecasts into B4:D8 without having to make any changes. 
For the Normal category, we only need to change the table of probabilities that we are looking up (and make sure that we point back to Regions!B4):
=HLOOKUP(Regions!B4,Probabilities!$B$11:$D$17,Forecasts!$A5-1997)

The new probabilities table for Normal is highlighted below indicating that we still need row 4 of the table to obtain the probabilities for 2001.

	
	A
	B
	C
	D

	10
	N
	
	
	

	11
	Region
	I
	II
	III

	12
	
	
	
	

	13
	
	
	
	

	14
	2001
	35%
	35%
	35%

	15
	2002
	35%
	35%
	40%

	16
	2003
	35%
	35%
	45%

	17
	2004
	40%
	35%
	45%


Similarly, adjust the table to lookup the Above-normal probabilities.

This HLOOKUP function call would be simplified if we used the forecast indicators discussed in footnote 2 rather than the regions. The probability table could be set up so that the three categories are indicated (of course, additional columns will be needed so that all the various combinations of probabilities are covered):

Indicators
	
	A
	B
	C
	D
	E

	1
	Indicator
	I
	II
	III
	IV

	2
	B
	25%
	20%
	40%
	45%

	3
	N
	35%
	35%
	35%
	35%

	4
	A
	40%
	45%
	25%
	20%


The HLOOKUP call then becomes:

=HLOOKUP(Regions!B4,Probabilities!$B$1:$E$4,2)

Change the last argument to 3 for Normal and 4 for Above-normal.

Unconditional bias
To calculate the unconditional biases take each category separately, and average the forecast probabilities for all years at each station, and then calculate how frequently each category did occur. Next average across all the stations, and subtract the observed relative frequency from the average forecasts.
To calculate the average probabilities, work from the Forecasts worksheet and immediately below the last forecast for Below use 

=AVERAGE(B5:B8)

Copy the formula for all stations; it will automatically update the cell references. The formula can also be copied immediately below the last forecasts for Normal and for Above. To average across the stations use

=AVERAGE(B9:D9)

Linear Probability Scores
Calculate the LPS first because we can use the results of the calculations for other scores. The LPS can be obtained by using

=IF(Category!B4="B",Forecasts!B5,
IF(Category!B4="N",Forecasts!B14,Forecasts!B23))

If there are missing values, an additional check may be required for Cateogry!B4=”A”). Enter this formula in a new worksheet LPS. It can be copied to obtain the probability on the verifying category for all the years and locations.

LPS

	
	A
	B
	C
	D

	1
	Station
	Station A
	Station B
	Station C

	2
	Latitude
	6.25
	8.75
	11.25

	3
	Longitude
	1.25
	1.25
	1.25

	4
	2001
	25%
	25%
	25%

	5
	2002
	35%
	35%
	35%

	6
	2003
	40%
	40%
	40%

	7
	2004
	30%
	35%
	30%

	8
	Average
	33%
	34%
	33%

	9
	Score
	33%
	
	


Calculate the score for each station using AVERAGE, and then average over all the cases. It may be also of interest to average across the stations for each year, and then to plot a graph of the score against time.
Ranked Hits
If we calculate whether each verifying category had the highest, second highest, or lowest probability, we can then count up how many there were in each rank. In Excel 2010 the RANK.AVG command can be used:

=RANK.AVG(LPS!B4,(Forecasts!B5,Forecasts!B14,Forecasts!B23)))

This function identifies whether the probability on the verifying category (which we determined previously in LPS!B4), is the highest of the probabilities for the three categories, given in Forecast!B5, Forecast!B14, and Forecast!B23. It returns a 1 if the verifying category is the highest, a 2 if it is the second highest and a 3 if it is the lowest. If the verifying category had equal highest probability (for example, if the forecast is B=40, N=40, and A=20, and B occurs), then this probability is tied first and second highest, and the function will return the average of the ranks, which is 1.5. Similarly, if the verifying category had equal lowest probability (for example, if the forecast is B=30, N=40, and A=30, and B occurs), then this probability is tied second and third highest, and the function will return the average of these ranks, which is 2.5. However, this averaging of the ranks creates a problem in cases when the forecast is for climatological probabilities. In that case the verifying observation will be in the category with tied first, second, and third highest probability, and the average of those ranks is 2. We will therefore be unable to distinguish a case of the observation verifying in the category with the second highest probability from a climatological probability. To avoid this problem, imbed the command above in an IF statement so that cases of climatological forecasts are flagged as -1:
=IF(LPS!B4=1/3,-1,RANK.AVG(LPS!B4,(Forecasts!B5,Forecasts!B14,Forecasts!B23)))

This formula can be copied to obtain the rank for each year and location. We can then calculate averages over time and or space, as with the LPS.
In older versions of Excel, the RANK.AVG function is unavailable, and RANK does not test for whether there are ties. As a result, a tied equal highest probability will return a 1 rather than a 1.5. We can work around this problem by first ranking in descending order (the default; replace RANK.AVG with RANK in the formula above) and then setting up a separate table with the ranks in ascending order:
=IF(LPS!B4=1/3,-1,RANK(LPS!B4,(Forecasts!B5,Forecasts!B14,Forecasts!B23),1))

In this second table, if the verifying category has the highest probability it will score a 3, but it will score a 2 if the probability is tied equal highest. We can convert these ascending ranks to descending ranks by applying the formula:

descending_rank = 4 – ascending_rank

If we then average this result with the ascending ranks from Excel’s RANK function, we obtain the ranks we need with the ties correctly identified:

descending_tied_rank = (descending_rank + 4 – ascending_rank) / 2
descending_tied_rank = 2 + (descending_rank – ascending_rank) / 2
Ranks_old
	
	A
	B
	C
	D

	1
	Descending
	
	
	

	2
	Station
	Station A
	Station B
	Station C

	3
	Latitude
	6.25
	8.75
	11.25

	4
	Longitude
	1.25
	1.25
	1.25

	5
	2001
	3
	3
	3

	6
	2002
	2
	2
	2

	7
	2003
	1
	1
	1

	8
	2004
	2
	2
	2

	9
	
	
	
	

	10
	Ascending
	
	
	

	11
	Station
	Station A
	Station B
	Station C

	12
	Latitude
	6.25
	8.75
	11.25

	13
	Longitude
	1.25
	1.25
	1.25

	14
	2001
	1
	1
	1

	15
	2002
	2
	2
	2

	16
	2003
	1
	1
	1

	17
	2004
	1
	2
	1

	18
	
	
	
	

	19
	Ties
	
	
	

	20
	Station
	Station A
	Station B
	Station C

	21
	Latitude
	6.25
	8.75
	11.25

	22
	Longitude
	1.25
	1.25
	1.25

	23
	2001
	3
	3
	3

	24
	2002
	2
	2
	2

	25
	2003
	1
	1
	1

	26
	2004
	2.5
	2
	2


Given the tables shown in Hits_old, the corrected rankings could be obtained entering the following formula into B23, and copying it through D26:
=IF(B5=-1,-1,2+(B5-B14)/2)
Profits and Losses
The profits or losses made by investing on the forecasts given fair odds are 2 ( the probability on the verifying category ( the investment made. Start a new worksheet called Profits, and include an additional row for the year before the forecasts start. In this row, shown by the year 2000, enter an initial investment for each location. Rather than just calculating the profits or losses it is simpler to calculate the balance: after the first year the balance is the initial investment plus or minus the profits or losses, and so can be calculated simply by multiplying the previous year’s balance by 3 ( the probability on the verifying category. So in cell B5 enter:
=B4*LPS!B4*3
This formula can be copied to complete the table.
Profits
	
	A
	B
	C
	D
	E
	F

	1
	Station
	Station A
	Station B
	Station C
	
	Region-wide

	2
	Latitude
	6.25
	8.75
	11.25
	
	

	3
	Longitude
	1.25
	1.25
	1.25
	
	

	4
	2000
	$100.00
	$100.00
	$100.00
	
	$300.00

	5
	2001
	$75.00
	$75.00
	$75.00
	
	$225.00

	6
	2002
	$78.75
	$78.75
	$78.75
	
	$236.25

	7
	2003
	$94.50
	$94.50
	$94.50
	
	$283.50

	8
	2004
	$85.05
	$93.23
	$85.05
	
	$269.33

	9
	
	
	
	
	
	

	10
	Interest
	-3.97%
	-0.19%
	-3.97%
	
	-2.66%


The effective annual interest rate can be calculated by entering the following formula into B9:
=(B8/B4)^(1/COUNT(B5:B8))-1

Region-wide results can be obtained by summing across the columns. In cell F4 enter
=SUM(B4:D4)

and copy it into the subsequent rows. The formula for the interest rate calculation can be copied from B9 into F9. 
Reliability Diagrams
Set up a table on a new worksheet, Reliability, with all the possible forecast probabilities in column A (i.e., from 0% to 100% at intervals of 5%, but with an additional row for 33%). 

Reliability
	
	A
	B
	C
	D
	E

	1
	B
	
	
	
	

	2
	Forecasts
	Frequency
	Rel. Freq.
	Hits
	Hit score

	3
	0%
	
	
	
	

	4
	5%
	
	
	
	

	5
	10%
	
	
	
	

	6
	15%
	
	
	
	

	7
	20%
	
	
	
	

	8
	25%
	
	
	
	

	9
	30%
	
	
	
	

	10
	33%
	
	
	
	

	11
	35%
	
	
	
	

	12
	40%
	
	
	
	

	13
	45%
	
	
	
	

	14
	50%
	
	
	
	

	15
	55%
	
	
	
	

	16
	60%
	
	
	
	

	17
	65%
	
	
	
	

	18
	70%
	
	
	
	

	19
	75%
	
	
	
	

	20
	80%
	
	
	
	

	21
	85%
	
	
	
	

	22
	90%
	
	
	
	

	23
	95%
	
	
	
	

	24
	100%
	
	
	
	

	25
	
	
	
	
	

	26
	
	
	
	
	


Calculate the frequency with which each forecast probability is issued: in B3 enter

=COUNTIF(Forecasts!$B$5:$D$8,"="&A3)

This formula can be copied through B24, and converted to a relative frequency for plotting a histogram by entering into C3
=B3/SUM(B$3:B$24)
Now calculate for how many of those forecasts there was an event. Combine the COUNTIF statement with an additional test to see if rainfall was Below. Excel allows us to combine multiple conditional statements using COUNTIFS
. In D3 enter

=COUNTIFS(Forecasts!$B$5:$D$8,"="&A3,Categories!$B$4:$D$7,”=B”)

Copy this formula through D24. To calculate the observed relative frequency (which could be interpreted as a hit score) for each forecast probability, divide the number of hits (column D) by the number of forecasts (column B). However, to avoid dividing by zero if there were no forecasts of a given probability, embed this calculation in an IF statement: enter into cell E3
=IF(B3>0,D3/B3,NA())

It is now possible to construct a graph using column A as the x-axis, and columns E and C as y-axes.
Reliability diagrams for Normal and Above can be obtained by modifying the calculations in column B so that the table of the respective forecast probabilities is accessed
=COUNTIF(Forecasts!$B$14:$D$17,"="&A3)
=COUNTIF(Forecasts!$B$23:$D$26,"="&A3)

(for Normal and Above, respectively). Similarly, the calculations in column D need to be modified:

=COUNTIFS(Forecasts!$B$14:$D$17,"="&A3,Categories!$B$4:$D$7,”=N”)
=COUNTIFS(Forecasts!$B$23:$D$26,"="&A3,Categories!$B$4:$D$7,”=A”)

� The italics represent the name of the worksheet.


� In Spanish, use a semi-colon separator instead of a comma, and the command is =PERCENTIL(B$4:B$33;1/3). In French the semi-colon separator is also required, and the command is =CENTILE(B$4:B$33;1/3).


� In Spanish: =CONTAR.SI(B$4:B$33;"<"&B52)/CONTAR(B$4:B$33).�In French: =NB.SI(B$4:B$33;"<"&B52)/COUNT(B$4:B$33).


� In Spanish and French:�=SI(Observations!B44<=Observations!B$52;"B"; SI(Observations!B44<=Observations!B$53;"N";"A"))


� To demonstrate that the regions do not matter, instead of using region numbers, we could assign each unique set of probabilities a number. For example, forecast I represents B=40, N=35, A=25; forecast II represents B=45, N=35, A=20, etc. We could then replace the table above we one that contained the definitions, and since the definitions would be the same every year, we could just have one row for each category.


� In Spanish =BUSCARH(Regions!B4;Probabilities!$B$2:$D$6:Forecasts!$A5-1997).�In French: =RECHERCHEH(Regions!B4;Probabilities!$B$2:$D$6:Forecasts!$A5-1997) .





� In Spanish. PROMEDIO. In French, MOYENNE.


� In Spanish and French: =SI(=IF(Category!B4= "B";Forecasts!B5;SI(Category!B4= "N";Forecasts!B14;Forecasts!B23))


� In Spanish: =JERARQUIA.MEDIA(LPS!B4;(Forecasts!B5;Forecasts!B14;Forecasts!B23))).�In French: =MOYENNE.RANG(LPS!B4;(Forecasts!B5;Forecasts!B14;Forecasts!B23)))


� In Spanish:  =SUMA(B4:D4). In French: =SOMME(B4:D4). 


� In Spanish: CONTAR.SI.CONUNTO. In French: NB.SI.ENS.





